
Parametric Curves  

> restart;  

We begin with a simple plot of a parametric curve.  

Its parametric equations are  

> x:=t->cos(t)+cos(7*t)/2+sin(17*t)/3; 
y:=t->sin(t)+sin(7*t)/2+cos(17*t)/3;  

  

  

To plot a parametric curve we can use the plot command as before.  

However, the syntax is slightly different from the one used for plotting functions.  

We now need to specify the coordinate functions along with a range for the parameter   

(which we take as  in this example) all enclosed in square brackets.  

> plot([x(t), y(t), t=0..2*Pi]);  

  

Next we illustrate the generation of a cycloid .  
From class we know that a cycloid is obtained by tracking the position of a fixed point on the 

circumference of a circle with radius  rolling along a straight line.  

The parametric equations are  

> x:=t->r*(t-sin(t)); 
y:=t->r*(1-cos(t));  

  

  

We will generate the picture in several steps.  

First we use the Maple command animate to create an animation of a circle with radius  moving 

along the  -axis from 0 to   

(we choose  to see two full arches of the cycloid).  

Animate is part of the plots package which has to be loaded first.  

> with(plots):  

To have a specific value for the radius r of the circle for the plot we set r=1.  

> r:=1:  

Note that the animate command requires as parameters  

- a curve to be drawn,  

- a second parameter  (the frame parameter ) which will simulate the movement of the circle from 

left to right in our example.  

- the definition of the number of frames to be generated.  

One can think of animate as a command which generates a movie .  

The result of animate is a sequence of frames, which when viewed successively at a high enough speed 
look like a movie.  

Here is the circle:  

(To start the animation you must click on the plot and then  

- either use the VCR buttons which appear at the top of the screen,  

- select Play from the Animation pull down menu at the top of the screen,  

- or click on the plot with the right mouse button and choose animate-play.)  

> animate( [-r*sin(t) + d, -r*cos(t) + r, t=0..2*Pi], d=0..4*r*Pi, frames=30, color=magenta, 
scaling=constrained, axes=frame, view=[-Pi/2..9*Pi/2,0..2]);  

  

Let's take a closer look at how animate works, and what its parameters are.  

First of all, you can disregard the options color , scaling , axes , and view . They are for cosmetic 
purposes only.  

As noted above, the option frames tells Maple how many individual frames to produce.  

Let's use a trimmed down version of the above command to make things a little more transparent:  

> animate( [-r*sin(t) + d, -r*cos(t) + r, t=0..2*Pi], d=0..4*r*Pi, frames=3);  

If you animated the previous plot you probably noticed that the animation was not very smooth (there 
weren't enough frames).  

But that should not bother us right now.  

Let's try to understand the other two parameters remaining:  

- the curve  

- and frame parameter  .  

Notice that the first argument in animate is a curve  

(here a parametric curve, although other objects are also possible),  

and it is generated (for any fixed value of the frame parameter  ) in our example by the curve 

parameter  .  

So, for  , say, the above command does the same as the following plot statement, i.e. it just draws 

a circle:  

> f1:=plot( [-r*sin(t) + 0, -r*cos(t) + r, t=0..2*Pi] ): 
display(f1);  

Now we can take another value, e.g.  , and we get the same circle - but shifted  units to 

the right.  

This is the second frame generated in our simplified animate command above.  

> f2:=plot( [-r*sin(t) + 2*Pi, -r*cos(t) + r, t=0..2*Pi] ): 
display(f2);  

  

Finally, we can produce a third plot, with  :  

> f3:=plot( [-r*sin(t) + 4*Pi, -r*cos(t) + r, t=0..2*Pi] ): 
display(f3);  

And all that animate does, is it takes these three plots and displays them in sequence.  

The following use of display accomplishes the same thing:  

> display([f1,f2,f3], insequence=true);  

  

Now we return to the cycloid.  

We save our animation of the rolling circle in the variable a1 so that we can later display it together with 
the cycloid.  

> a1 := animate( [-r*sin(t) + d, -r*cos(t) + r, t=0..2*Pi], d=0..4*r*Pi, frames=30, color=magenta, 

scaling=constrained, axes=frame, view=[-Pi/2..9*Pi/2,0..2]):  

Next we create an animation of the moving point on the circumference of the circle by creating a little 

blue tickmark which rotates along with the circle.  

We save this part in a2.  

> a2 := animate( [-r*t*sin(d) + d, -r*t*cos(d)+r, t=0.9..1.1], d=0..4*r*Pi, frames=30, color=blue, 

scaling=constrained, axes=frame, view=[-Pi/2..9*Pi/2,0..2]):  

Finally, we create the animation for the cycloid curve, and save it in a3.  

Note how the frame parameter is used here:  

The graph of the cycloid is given by  

> plot([x(t), y(t), t=0..4*Pi], scaling=constrained, color=green);  

  

In order to simulate the curve being drawn we use the frame parameter  to determine how much of 

the curve we want drawn.  

So, in the statement to follow the range of  is  , with  

 corresponding to "nothing is drawn",  

 "the entire curve is drawn", 

0<  <1 "a fraction of the curve is drawn".  

> a3 := animate( [x(d*t), y(d*t), t=0..4*Pi], d=0..1, frames=30, color=green, scaling=constrained, 

axes=frame, view=[-Pi/2..9*Pi/2,0..2]):  

Now we are ready to display all three animations (the rolling circle, the tickmark, and the curve) 
together.  

Note: you can modify the speed of the animation by clicking on the plot and then  
- either using the VCR buttons which appear at the top of the screen (>> makes the animation faster, << 
slower),  

- selecting Faster/Slower from the Animation pull down menu at the top of the screen,  
- or clicking on the plot with the right mouse button and choose animate-faster/slower.  

> display({a1,a2,a3});  

  

The following creates an animation of a hypocycloid (the curve obtained by following the motion of a 

fixed point on the circumference of a circle with radius  which rolls along the inside of another circle 

with radius  >  ):  

The general parametric equations are:  

> x:=t->(a-b)*cos(t)+b*cos((a-b)*t/b); 
y:=t->(a-b)*sin(t)-b*sin((a-b)*t/b); 

  

  

For the following plot we choose  ,  :  

> a:=4: b:=1: 
a1:=animate([(a-b)*cos(d)+b*cos(t), (a-b)*sin(d)+b*sin(t), t=0..2*Pi], d=0..2*Pi, frames=30, 
color=magenta, view=[-a..a,-a..a]): # animation of the rolling circle 

a2:=plot([a*cos(t), a*sin(t), t=0..2*Pi], color=red): # plot of the outer circle 
a3:=animate([x(d*t), y(d*t), t=0..2*Pi], d=0..1, frames=30, color=green, view=[-a..a,-a..a]): # 
animation of the hypocycloid 
a4:=animate([t*x(d), t*y(d), t=.98..1.02], d=0..2*Pi, frames=30, color=blue, view=[-a..a,-a..a]): # 
animation of the tickmark 
display({a1,a2,a3,a4}); 
a:='a': b:='b':  

  

An epicycloid is a curve obtained by following the motion of a fixed point on the circumference of a 

circle with radius  which rolls along the outside of another circle with radius  >  .  

Its general parametric equations are given by:  

> x:=t->(a+b)*cos(t)-b*cos((a+b)*t/b); 
y:=t->(a+b)*sin(t)-b*sin((a+b)*t/b);  

  

  

And here is the animation of an epicycloid with  ,  :  

> a:=4: b:=1: 
a1:=animate([(a+b)*cos(d)+b*cos(t), (a+b)*sin(d)+b*sin(t), t=0..2*Pi], d=0..2*Pi, frames=30, 
color=magenta, view=[-(a+2*b)..(a+2*b),-(a+2*b)..(a+2*b)]): # animation of the rolling circle 
a2:=plot([a*cos(t), a*sin(t), t=0..2*Pi], color=red): # plot of the inner circle 
a3:=animate([x(d*t), y(d*t), t=0..2*Pi], d=0..1, frames=30, color=green, view=[-(a+2*b)..(a+2*b),-
(a+2*b)..(a+2*b)]): # animation of the epicycloid 

a4:=animate([t*x(d), t*y(d), t=.98..1.02], d=0..2*Pi, frames=30, color=blue, view=[-(a+2*b)..
(a+2*b),-(a+2*b)..(a+2*b)]): # animation of the tickmark 
display({a1,a2,a3,a4});  

  

Assignment:  

1. a) Create a plot of the following parametric curve:  

 ,  

 ,  

 in  .  

b) What happens if you replace 2 by -2?  

c) What happens if you replace 2 by 3?  

d) What happens as you increase the value even further?  

2. What is the difference between the curves  

 ,  

 ,  

 in  ,  

and  

 ,  

 ,  

 in  ?  

3. Lissajous curves are the type of curves that appear on oscilloscopes in physics or electronics. They 
are given by  

 ,  

 .  

a) Plot the case  ,  in  .  

b) What happens for  ?  

c) Create an animation that shows how the curve in a) is being drawn.  

4. Let  

  

  

a) Plot the "slinky curve" corresponding to  with  in 

 .  

b) Play with the parameters to create an even prettier (more interesting) curve. Be creative!  

5. Consider the parametric curve  ,  ,  in  . 

a) Plot the curve. 

b) Find  and  at the point  . 

c) Find an equation for the tangent line to the curve at  .  

d) Plot the curve together with the tangent line. 
e) Find the length of the curve. You will have to numerically evaluate the integral with evalf.  

